module - computing Volume ∇

In[1]:=
```
DumpGet["C:\\Users\\aliha\\Desktop\\wolfram-vertex-3D\\PREVIOUS
    CODE - slow heuns\\meshGen_noise.mx"];
```

In[2]:=
```
yLim[[1]] = 0.;
edges = SetPrecision[edges, 10];
faceListCoords = SetPrecision[faceListCoords, 10];
(*convert faceListCoords into an association*)
indToPtsAssoc = SetPrecision[indToPtsAssoc, 10];
ptsToIndAssoc = KeyMap[SetPrecision[#, 10] &, ptsToIndAssoc];
xLim = SetPrecision[xLim, 10];
yLim = SetPrecision[yLim, 10];
faceListCoords = Map[Lookup[indToPtsAssoc, #] &, cellVertexGrouping, {2}];
```

In[80]:=
```
Clear@periodicRules;
With[{xlim1 = xLim[[1]], xlim2 = xLim[[2]], ylim1 = yLim[[1]], ylim2 = yLim[[2]]},
  periodicRules = Dispatch[{
     {x_ /; x ≥ xlim2, y_ /; y ≤ ylim1, z_} :> SetPrecision[{x - xlim2, y + ylim2, z}, 10],
     {x_ /; x ≥ xlim2, y_ /; ylim1 < y < ylim2, z_} :> SetPrecision[{x - xlim2, y, z}, 10],
     {x_ /; xlim1 < x < xlim2, y_ /; y ≤ ylim1, z_} :> SetPrecision[{x, y + ylim2, z}, 10],
     {x_ /; x ≤ xlim1, y_ /; y ≤ ylim1, z_} :> SetPrecision[{x + xlim2, y + ylim2, z}, 10],
     {x_ /; x ≤ xlim1, y_ /; ylim1 < y < ylim2, z_} :> SetPrecision[{x + xlim2, y, z}, 10],
     {x_ /; x ≤ xlim1, y_ /; y ≥ ylim2, z_} :> SetPrecision[{x + xlim2, y - ylim2, z}, 10],
     {x_ /; xlim1 < x < xlim2, y_ /; y ≥ ylim2, z_} :> SetPrecision[{x, y - ylim2, z}, 10],
     {x_ /; x ≥ xlim2, y_ /; y ≥ ylim2, z_} :> SetPrecision[{x - xlim2, y - ylim2, z}, 10]
    }];
  transformRules = Dispatch[{
     {x_ /; x ≥ xlim2, y_ /; y ≤ ylim1, _} :> SetPrecision[{-xlim2, ylim2, 0}, 10],
     {x_ /; x ≥ xlim2, y_ /; ylim1 < y < ylim2, _} :> SetPrecision[{-xlim2, 0, 0}, 10],
     {x_ /; xlim1 < x < xlim2, y_ /; y ≤ ylim1, _} :> SetPrecision[{0, ylim2, 0}, 10],
     {x_ /; x ≤ xlim1, y_ /; y ≤ ylim1, _} :> SetPrecision[{xlim2, ylim2, 0}, 10],
     {x_ /; x ≤ xlim1, y_ /; ylim1 < y < ylim2, _} :> SetPrecision[{xlim2, 0, 0}, 10],
     {x_ /; x ≤ xlim1, y_ /; y ≥ ylim2, _} :> SetPrecision[{xlim2, -ylim2, 0}, 10],
     {x_ /; xlim1 < x < xlim2, y_ /; y ≥ ylim2, _} :> SetPrecision[{0, -ylim2, 0}, 10],
     {x_ /; x ≥ xlim2, y_ /; y ≥ ylim2, _} :> SetPrecision[{-xlim2, -ylim2, 0}, 10],
     {___Real} :> SetPrecision[{0, 0, 0}, 10]}];
 ];
```

In[167]:=
```
origcellOrient = <|MapIndexed[First[#2] → #1 &, faceListCoords]|>;
boundaryCells =
  With[{xlim1 = xLim[[1]], xlim2 = xLim[[2]], ylim1 = yLim[[1]], ylim2 = yLim[[2]]},
   Union[First /@ Position[origcellOrient,
       {x_ /; x ≥ xlim2, __} | {x_ /; x ≤ xlim1, __} |
        {_, y_ /; y ≥ ylim2, _} | {_, y_ /; y ≤ ylim1, _}] /. Key[x_] :→ x]
  ];
wrappedMat = AssociationThread[
   Keys[cellVertexGrouping] → Map[Lookup[indToPtsAssoc, #] /. periodicRules &,
     Lookup[cellVertexGrouping, Keys[cellVertexGrouping]], {2}]];
```

In[85]:=
```
meanTri = Compile[{{faces, _Real, 2}},
  Mean@faces,
  CompilationTarget → "C", RuntimeAttributes → {Listable},
  Parallelization → True
 ]
```

Out[85]= CompiledFunction[Argument count: 1 Argument types: {{_Real, 2}}]

In[86]:=
```
Clear[triNormal];
triNormal = Compile[{{ls, _Real, 2}},
  Block[{res},
   res = Partition[ls, 2, 1];
   Cross[res[[1, 1]] - res[[1, 2]], res[[2, 1]] - res[[2, 2]]]
  ], CompilationTarget → "C", RuntimeAttributes → {Listable}
 ]
```

Out[87]= CompiledFunction[Argument count: 1 Argument types: {{_Real, 2}}]

In[88]:=
```
Clear[meanFaces, triangulateToMesh];
meanFaces = Compile[{{faces, _Real, 2}},
  Block[{facepart, edgelen, mean},
   facepart = Partition[faces, 2, 1];
   AppendTo[facepart, {facepart[[-1, -1]], faces[[1]]}];
   edgelen = Table[Norm[SetPrecision[First@i - Last@i, 10]], {i, facepart}];
   mean = Total[edgelen * (Mean /@ facepart)] / Total[edgelen];
   mean],
  RuntimeAttributes → {Listable}, CompilationTarget → "C",
  CompilationOptions → {"InlineExternalDefinitions" → True}
 ]

triangulateToMesh[faces_] := Block[{mf, partfaces},
   mf = SetPrecision[meanFaces@faces, 10];
   partfaces = Partition[#, 2, 1, 1] & /@ faces;
   MapThread[
    If[Length[#] ≠ 3,
      Function[x, Join[x, {#2}]] /@ #1,
      {#[[All, 1]]}
     ] &, {partfaces, mf}]
  ];
```

Out[89]= CompiledFunction[Argument count: 1 Argument types: {{_Real, 2}}]

In[93]:=
```
Clear@cellCentroids;
cellCentroids[polyhedCentAssoc_, keystopo_, shiftvec_] :=
  Block[{assoc = <||>, regcent, counter},
   AssociationThread[Keys@keystopo →
     KeyValueMap[
      Function[{key, cellassoc},
       If[KeyFreeQ[shiftvec, key],
        Lookup[polyhedCentAssoc, cellassoc],
        If[KeyFreeQ[shiftvec[key], #],
            regcent = polyhedCentAssoc[#],
            regcent = polyhedCentAssoc[#] + shiftvec[key][#];
            regcent
           ] & /@ cellassoc
       ]
      ], keystopo]
    ]
  ];
```

In[95]:=
```
𝒟 = Rectangle[{First@xLim, First@yLim}, {Last@xLim, Last@yLim}];
```

In[96]:=
```
getLocalTopology[ptsToIndAssoc_, indToPtsAssoc_, vertexToCell_,
    cellVertexGrouping_, wrappedMat_, faceListCoords_][vertices_] :=
  Block[{localtopology = <||>, wrappedcellList = {}, vertcellconns,
```

```
localcellunion, v, wrappedcellpos, vertcs = vertices, rl1, rl2,
transVector, wrappedcellCoords, wrappedcells, vertOutofBounds,
shiftedPt, transvecList = {}, $faceListCoords = Values@faceListCoords,
vertexQ, boundsCheck, rules, extractcellkeys, vertind,
cellsconnected, wrappedcellsrem},
vertexQ = MatchQ[vertices, {__?NumberQ}];
If[vertexQ,
 (vertcellconns =
   AssociationThread[{#} , {vertexToCell[ptsToIndAssoc[#]]}] &@vertices;
  vertcs = {vertices};
  localcellunion = Flatten[Values@vertcellconns]),
 (vertcellconns = AssociationThread[# ,
      Lookup[vertexToCell, Lookup[ptsToIndAssoc, #]]] &@vertices;
  localcellunion = Union@Flatten[Values@vertcellconns])
];

If[localcellunion ≠ {},
 AppendTo[localtopology,
  Thread[localcellunion →
    Map[Lookup[indToPtsAssoc, #] &, cellVertexGrouping /@ localcellunion, {2}]]
 ]
];
(* condition to be an internal edge: both vertices should have 3 neighbours *)
(* if a vertex has 3 cells in its local neighbourhood then the entire
  network topology about the vertex is known → no wrapping required *)
(* else we need to wrap around the vertex because other cells
  are connected to it → periodic boundary conditions *)
With[{vert = #},
   vertind = ptsToIndAssoc[vert];
   cellsconnected = vertexToCell[vertind];
   If[Length[cellsconnected] ≠ 3,
    If[ (𝒟~RegionMember~Most[vert]),
      (*Print["vertex inside bounds"];*)
      v = vert;
      With[{x = v[[1]], y = v[[2]]}, boundsCheck =
        (x == xLim[[1]] || x == xLim[[2]] || y == yLim[[1]] || y == yLim[[2]])];

      extractcellkeys = If[boundsCheck,
        {rl1, rl2} = {v, v /. periodicRules};
        rules = Block[{x$},
          With[{r = rl1, s = rl2},
           DeleteDuplicates[HoldPattern[SameQ[x$, r]] || HoldPattern[SameQ[x$, s]]]
          ]
         ];
        Position @@ With[{rule = rules},
          Hold[wrappedMat, x_ /; ReleaseHold@rule, {3}]
         ],
        Position[wrappedMat, x_ /; SameQ[x, v], {3}]
       ];
      (* find cell indices that are attached to the vertex in wrappedMat *)
```

```
wrappedcellpos = DeleteDuplicatesBy[
  Cases[extractcellkeys,
   {Key[p : Except[Alternatives @@
         Join[localcellunion, Flatten@wrappedcellList]]], y__} :→ {p, y}],
  First];
(*wrappedcellpos = wrappedcellpos/.
   {Alternatives@@Flatten[wrappedcellList],__} :→ Sequence[];*)
(* if a wrapped cell has not been considered earlier (i.e. is new)
 then we translate it to the position of the vertex *)
If[wrappedcellpos ≠ {},
 If[vertexQ,
  transVector = SetPrecision[(v - Extract[$faceListCoords,
         Replace[#, {p_, q__} :→ {Key[p], q}, {1}]]) & /@ wrappedcellpos, 10],
  (* call to function is enquiring an edge and not a vertex*)
  transVector =
   SetPrecision[(v - Extract[$faceListCoords, #]) & /@ wrappedcellpos, 10]
 ];
 wrappedcellCoords = MapThread[#1 → Map[Function[x,
       SetPrecision[x + #2, 10]], $faceListCoords[[#1]], {2}] &,
   {First /@ wrappedcellpos, transVector}];
 wrappedcells = Keys@wrappedcellCoords;
 AppendTo[wrappedcellList, Flatten@wrappedcells];
 AppendTo[transvecList, transVector];
 AppendTo[localtopology, wrappedcellCoords];
],
(* the else clause: vertex is out of bounds *)
(*Print["vertex out of bounds"];*)
vertOutofBounds = vert;
(* translate the vertex back into mesh *)
transVector = vertOutofBounds /. transformRules;
shiftedPt = SetPrecision[vertOutofBounds + transVector, 10];
(* ------------- CORE B ------------- *)
(* find which cells the
 shifted vertex is a part of in the wrapped matrix *)
wrappedcells = Complement[
  Union@Cases[Position[wrappedMat, x_ /; SameQ[x, shiftedPt], {3}],
     x_Key :→ Sequence @@ x, {2}] /.
   Alternatives @@ localcellunion → Sequence[],
  Flatten@wrappedcellList];

(*forming local topology now that we know the wrapped cells *)
If[wrappedcells ≠ {},
 AppendTo[wrappedcellList, Flatten@wrappedcells];
 wrappedcellCoords = AssociationThread[wrappedcells,
   Map[Lookup[indToPtsAssoc, #] &,
    cellVertexGrouping[#] & /@ wrappedcells, {2}]];
 With[{opt = (vertOutofBounds /. periodicRules)},
  Block[{pos, vertref, transvec},
    Do[
     With[{cellcoords = wrappedcellCoords[cell]},
```

```
                pos = FirstPosition[cellcoords /. periodicRules, opt];
                vertref = Extract[cellcoords, pos];
                transvec = SetPrecision[vertOutofBounds - vertref, 10];
                AppendTo[transvecList, transvec];
                AppendTo[localtopology,
                 cell → Map[SetPrecision[# + transvec, 10] &, cellcoords, {2}]];
               ], {cell, wrappedcells}]
             ];
           ];
         ];
         (* to detect wrapped cells not detected by CORE B*)
         (* ------------- CORE C ------------- *)
         Block[{pos, celllocs, ls, transvec, assoc, tvecLs = {}, ckey},
          ls = Union@Flatten@Join[cellsconnected, wrappedcells];
          If[Length[ls] ≠ 3,
           pos = Position[faceListCoords, x_ /; SameQ[x, shiftedPt], {3}];
           celllocs = DeleteDuplicatesBy[Cases[pos, Except[{Key[Alternatives @@ ls],
                 __}]], First] /. {Key[x_], z__} ⧴ {Key[x], {z}};
           If[celllocs ≠ {},
            celllocs = Transpose@celllocs;
            assoc = <|
              MapThread[
               (transvec = SetPrecision[
                   vertOutofBounds - Extract[faceListCoords[Sequence @@ #1], #2], 10];
                 ckey = Identity @@ #1;
                 AppendTo[tvecLs, transvec];
                 ckey → Map[SetPrecision[Lookup[indToPtsAssoc, #] + transvec, 10] &,
                   cellVertexGrouping[Sequence @@ #1], {2}]
                ) &, celllocs]
              |>;
            AppendTo[localtopology, assoc];
            AppendTo[wrappedcellList, Keys@assoc];
            AppendTo[transvecList, tvecLs];
           ];
          ];
         ];
        ];
      ];
    ] & /@ vertcs;

  transvecList = Which[
    MatchQ[transvecList, {{{__?NumberQ}}}], First[transvecList],
    MatchQ[transvecList, {{__?NumberQ} ..}], transvecList,
    True, transvecList //. {x___, {p : {__?NumberQ} ..}, y___} ⧴ {x, p, y}
   ];
  {localtopology, Flatten@wrappedcellList, transvecList}
 ];
```

Launch Kernels

In[24]:=
```
LaunchKernels[]
```

Out[◦]=
```
{KernelObject[ Name: local KernelID: 1 ], KernelObject[ Name: local KernelID: 2 ],
 KernelObject[ Name: local KernelID: 3 ], KernelObject[ Name: local KernelID: 4 ]}
```

prerequisite run

In[97]:=
```
Graphics3D[Polygon /@ (faceListCoords /@ boundaryCells)]
```

Out[97]=



In[◦]:=
```
(*missing boundary cells need to be found *)
```

In[98]:=
```
bcells = KeyTake[faceListCoords, boundaryCells];
```

In[99]:=
```
Length@boundaryCells
```

Out[99]= 60

In[102]:=
```
keyLs = Union@(Flatten@Lookup[vertexToCell,
      Lookup[ptsToIndAssoc,
       With[{xlim1 = xLim[[1]], ylim1 = yLim[[1]], ylim2 = yLim[[2]], xlim2 = xLim[[2]]},
         DeleteDuplicates@Cases[bcells,
           {x_ /; x ≥ xlim2, __} | {x_ /; x ≤ xlim1, __} |
            {_, y_ /; y ≥ ylim2, _} | {_, y_ /; y ≤ ylim1, _}, {3}]
         ] /. periodicRules
       ]
      ]~Join~boundaryCells);
```

In[103]:=
```
Length[keyLs] - Length[boundaryCells]
```

Out[103]= 16

In[104]:=
```
border = faceListCoords /@ keyLs;
```

In[105]:=
```
Graphics3D[{Polygon /@ border}, ImageSize → Medium]
```

Out[105]=



In[106]:=
```
wrappedMatC = KeyTake[wrappedMat, keyLs];
```

In[107]:=
```
vertKeys = Keys@indToPtsAssoc;
```

In[108]:=
```
(
  topo = <|
     # → (getLocalTopology[ptsToIndAssoc, indToPtsAssoc, vertexToCell, cellVertexGrouping,
           wrappedMatC, faceListCoords][indToPtsAssoc[#]] // First) & /@ vertKeys
     |>;
 ) // AbsoluteTiming
```

Out[108]= {0.911847, Null}

Growing/Static cells

randomly select cells in the mesh to grow

In[109]:=
```
cellIds = Keys@cellVertexGrouping;
```

In[110]:=
```
fractionPopulation = 0.07;
growingcellIndices = RandomSample[cellIds, Round[fractionPopulation Length@cellIds]]
```

Out[111]= {333, 266, 376, 73, 1, 264, 274, 237, 208, 118, 340, 193, 251,
 96, 111, 94, 278, 336, 37, 115, 182, 51, 91, 257, 356, 325, 225, 141}

In[112]:=
```
nongrowingCellIndices = cellIds ~ Complement ~ growingcellIndices;
```

finding triangles connected to a vertex

In[113]:=
```
pointind = 1167;
```

In[114]:=
```
point = indToPtsAssoc@pointind;
```

In[115]:=
```
triangulatedCells = triangulateToMesh /@ faceListCoords;
polyhedraAssoc = Polyhedron@Flatten[#, 1] & /@ triangulatedCells;
```

In[119]:=
```
(trimesh = Map[triangulateToMesh, topo, {2}]); // AbsoluteTiming
```

Out[119]=
```
{2.01512, Null}
```

In[118]:=
```
examplevertToTri =
   GroupBy[Flatten[Values@trimesh[#], 2], MemberQ[indToPtsAssoc[#]]][True] &[
    1]; // AbsoluteTiming
```

Out[118]=
```
{0.0002879, Null}
```

In[124]:=
```
(examplevertToTri =
    GroupBy[Flatten[Values@trimesh[#], 2], MemberQ[indToPtsAssoc[#]]][True];
   Graphics3D[{{Opacity[0.15], Blue, Triangle /@ examplevertToTri},
     Red, PointSize[0.03], Point@indToPtsAssoc[#]},
    ImageSize → Small]
  ) &[RandomInteger[Max@Keys@indToPtsAssoc]]
```

Out[124]=



In[125]:=
```
(associatedtri = With[{ItoPA = indToPtsAssoc, tmesh = trimesh},
     AssociationThread[vertKeys, Function[vert, <|GroupBy[
             Flatten[#, 1], MemberQ[ItoPA[vert]]
            ][True] & /@ tmesh[vert]|>] /@ vertKeys]
    ];
 ) // AbsoluteTiming
```

Out[125]=
```
{0.559277, Null}
```

In[126]:=
```
SeedRandom[3];
Graphics3D[{Opacity[0.1], Blue, Triangle /@
   Flatten[Values@Values@RandomSample[associatedtri, 30], 2]}, ImageSize → Large]
```

Out[127]=



In[128]:=
```
(centTri = <|# → meanTri[Values[associatedtri@#]] & /@ Keys@indToPtsAssoc|>;) //
 AbsoluteTiming
```

Out[128]= {0.393809, Null}

In[129]:=
```
centTri = SetPrecision[#, 10] & /@ centTri;
```

In[130]:=
```
(normals = Map[SetPrecision[#, 10] &, triNormal@Values@# & /@ associatedtri]); //
 AbsoluteTiming
```

Out[130]= {0.570228, Null}

In[131]:=
```
(normNormals = Map[Normalize, normals, {3}];) // AbsoluteTiming
```

Out[131]= {0.116208, Null}

In[132]:=
```
(triangulatedmesh = triangulateToMesh /@ faceListCoords); // AbsoluteTiming
(polyhedra = Polyhedron@* (Flatten[#, 1] &) /@ triangulatedmesh;) // AbsoluteTiming
```

Out[132]= {0.152934, Null}

Out[133]= {0.0153853, Null}

In[134]:=
```
(polyhedcent = RegionCentroid /@ polyhedra); // AbsoluteTiming
```

Out[134]= {4.21603, Null}

```
In[135]:= (
   topoF = <|
      # → (getLocalTopology[ptsToIndAssoc, indToPtsAssoc, vertexToCell, cellVertexGrouping,
            wrappedMatC, faceListCoords][indToPtsAssoc[#]]) & /@ vertKeys
      |>;
  ) // AbsoluteTiming

Out[135]= {0.94444, Null}
```

```
In[136]:= (keyslocaltopoF = Keys@*First /@ topoF); // AbsoluteTiming

Out[136]= {0.0047615, Null}
```

```
In[137]:= (shiftVecAssoc = Association /@ Map[Apply[Rule],
        Thread /@ Select[(#[[2 ;; 3]]) & /@ topoF, # ≠ {{}, {}} &], {2}]); // AbsoluteTiming

Out[137]= {0.0056096, Null}
```

```
In[138]:= (cellcentroids = cellCentroids[polyhedcent, keyslocaltopoF, shiftVecAssoc]);
```

```
In[139]:= (signednormals = AssociationThread[Keys@indToPtsAssoc,
       Map[
        MapThread[
          #2 Sign@MapThread[Function[{x, y}, (y - #1).x], {#2, #3}] &,
          {cellcentroids[#], normNormals[#], centTri[#]}] &, Keys@indToPtsAssoc]
       ]
   ); // AbsoluteTiming

Out[139]= {0.23582, Null}
```

make sets of open/closed triangles

```
In[140]:= opencloseTri = Flatten[Values@#, 1] & /@ associatedtri;
```

```
In[141]:= Graphics3D /@ Map[Triangle,
   GroupBy[GatherBy[opencloseTri[1], Intersection], Length, Flatten[#, 1] &], {2}]
```

Out[141]= ⟨| 1 → , 2 →  |⟩

In[142]:=
```
triDistAssoc = Block[{trianglemembers},
   Map[
    (trianglemembers = #;
      GroupBy[GatherBy[trianglemembers, Intersection], Length, Flatten[#, 1] &]) &,
    opencloseTri]
  ];
```

In[143]:=
```
{opentri, closedtri} = {triDistAssoc[pointind][1], triDistAssoc[pointind][2]};
```

In[144]:=
```
Graphics3D[{{Opacity[0.2], Red, Map[Triangle][opentri], Blue, Map[Triangle][closedtri]},
  {Black, PointSize[0.04], Point@indToPtsAssoc[pointind]}}, ImageSize → Small]
```

Out[144]=



associate normals with triangles

In[145]:=
```
vertTriNormalpairings = <|
   # → <|Thread[Flatten[Values@associatedtri[#], 1] → Flatten[signednormals@#, 1]]|> & /@
    vertKeys|>;
```

To associate the open/closed triangles with their respective normals we simply need to perform a lookup in the association for (vertex1,vertex2,vertex3) - a triangle face - and its normal.

In[146]:=
```
normalsO = Lookup[vertTriNormalpairings[pointind], opentri];
```

In[147]:=
```
normalsC = Lookup[vertTriNormalpairings[pointind], closedtri];
```

In[148]:=
```
normalLs = normalsO~Join~normalsC;
```

volume gradient F[x]

gradient of volume is computed as: $1/3 \sum A_\Delta \vec{N}$

In[149]:=
```
volumeGradient[point_, opentri_, closedtri_, normalLs_, cellids_,
    localtopology_, polyhedraAssoc_, growingCellIds_] :=
  Reap@Block[{topo, topology, normalassoc, gradV, gradVCont,
      triangulatedCellsSel, polyhedraSel, volume, growingIndkeys},
    triangulatedCellsSel = triangulateToMesh /@ localtopology;
    polyhedraSel = Lookup[polyhedraAssoc, cellids];
    topo =
     (Cases[#, x_ /; MemberQ[x, point]] &) @* (Flatten[#, 1] &) /@ triangulatedCellsSel;
    Sow[topo];
    normalassoc = AssociationThread[opentri ~ Join ~ closedtri, normalLs];
    gradV = Table[topology = topo[cell];
      (1.0 / 3.0)
       Total[Map[Area[Triangle[#]] * normalassoc[#] &, topology]], {cell, cellids}];
    volume = AssociationThread[cellids → ConstantArray[Vo, Length@cellids]];
    growingIndkeys =
     Replace[Intersection[cellids, growingCellIds], k_Integer :> {Key[k]}, {1}];
    volume = If[growingIndkeys ≠ {},
      Values@MapAt[ (1 + Vgrowth time) # &, volume, growingIndkeys],
      Values@volume
     ];
    gradVCont = kcv Total[ (Volume[polyhedraSel] / volume - 1) gradV]
   ];
```

In[150]:=
```
{cellids, localtopology} = Through[{Keys, Identity}[#]] &[First@topoF[pointind]];
```

In[151]:=
```
{res, pyramids} = volumeGradient[point, opentri, closedtri,
   normalLs, cellids, localtopology, polyhedraAssoc, growingcellIndices];
```

In[155]:=
```
Column@Through[{Length, Identity}[#]] &[res]
```

Out[155]=

3

$$\left\{k_{cv}\left(0.11384\left(-1+\frac{0.506575}{V_o}\right)-0.0956144\left(-1+\frac{0.588289}{V_o}\right)-0.0668358\left(-1+\frac{0.624122}{V_o}\right)\right),\right.$$
$$k_{cv}\left(-0.000830056\left(-1+\frac{0.506575}{V_o}\right)+0.11885\left(-1+\frac{0.588289}{V_o}\right)-0.123506\left(-1+\frac{0.624122}{V_o}\right)\right),$$
$$\left.k_{cv}\left(0.0482789\left(-1+\frac{0.506575}{V_o}\right)+0.0784126\left(-1+\frac{0.588289}{V_o}\right)+0.0646893\left(-1+\frac{0.624122}{V_o}\right)\right)\right\}$$

In[165]:=
```
plt = Graphics3D[{{Opacity[0.7], #}, {Black, PointSize[0.075], Point@point}},
    ImageSize → Tiny] & /@ Flatten@MapThread[Function[x, Tetrahedron@Join[{#}, x]] /@ #2 &,
    {cellcentroids[pointind], Values@pyramids[[1, 1]]}]
```

Out[165]= {
,
,
,
,
,
,
,
,
,
,
,
,
,
,
,
,
,

}

In[166]:=
```
Show[plt, Graphics3D[{Blue, PointSize[0.04], Point@cellcentroids[pointind]}],
 ImageSize → Small]
```

Out[166]=